

A talaj humuszanyagai természetes anyagok, növényi bomlástermékek, a környezetre nem mérgezők (a természetes vizekben 0,5–5 mg/L mennyiségben lehetnek jelen) – de a vízben oldódott anyagokkal kölcsönhatásba lépve azokká válhatnak. A huminanyagok több karboxil-, amino-, fenolos hidroxil-csoportot tartalmaznak, s ezekkel a nehézfém-ionokat kelát-komplex formában megkötik, és így biztosítják azok oldatban maradását. Ennek a folyamatnak a során valósul meg a fémek bioakkumulációja a vízi szervezetekben, minek eredményeként a nehézfém mennyisége ezekben több nagyságrenddel nagyobbá válhat (pl. halak húzában, kagylókban), mint amennyi volt eredetileg a vízben. A fémek oldhatósága nagymértékben függ a víz kémhatásától, mivel redoxpotenciáljuk a pH függvénye. Így az ólom a mérgező  $Pb^{2+}$  -ion formában van jelen, ha a vizes oldat pH-ja kisebb mint 6, 6 és 9 közti értéknél rosszul oldódó  $PbCO_3$  formában, de ha a víz redoxpotenciálja alacsonyabb a természetes értéknél, akkor elemi állapotban van. Ugyanakkor a víz biológiai szennyezettsége (mikroorganizmus tartalma) is befolyásolja a fémek oldódását. Például a fémes higanyról tudott, hogy vízben gyakorlatilag oldhatatlan. Amennyiben szervesanyag tartalmú a víz, a biológiai folyamatokat katalizáló mikroorganizmusok képesek metilezni a higanyt, miközben az metilhigannyá ( $CH_3Hg$ ), illetve dimetil-higannyá ( $(CH_3)_2Hg$ ) alakul (ezek mennyiségi aránya is pH függő, ha a  $pH < 8$ , akkor metil-higany, ha  $pH > 8$ , akkor dimetil-higany képződik nagyobb mennyiségben), amelyek a biológiai rendszerekbe már be tudnak épülni, a táplálékláncon keresztül az emberbe is bekerülnek. Veszélyes, mérgező hatású vegyületek.

#### Felhasznált könyvészeti anyag

1] E.Bârsan: Alimentări cu apă, Ed. Cermin, Iași, 2001

Máthé Enikő

## Érdekes informatika feladatok

XVIII. rész

#### Holdfázisok kiszámítása

A Hold Földünk természetes kísérője. Pályája enyhén lapult ellipszis, földközépen 354 000 km-re, földtávolban 404 000 ezer km-re van, átlagos távolsága pedig 384 000 km.

A Hold 27,3 nap alatt kerüli meg a Földet. Keringésének időtartama pontosan egybeesik tengelyforgásának időtartamával, ezért mindig ugyanazt az oldalt látjuk a Földről. Ezt nevezzük *Földre vonatkoztatott kötött tengelyforgásnak*.

A Holdon ugyanúgy váltakoznak a nappalok és az éjszakák, mint a Földön, de egy holdi nap hosszabb az előbb említett 27,3 napos időtartamnál – ennek az az oka, hogy a Föld kering a Nap körül, így változik a Nap csillagokhoz viszonyított helyzete. Ahhoz, hogy ezt az elmozdulást kompenzálja, a Holdnak kicsit tovább kell haladnia a pályáján (ehhez 2,2 napra van szüksége a Holdnak). Egy holdi nap időtartama így körülbelül 29,5 nap. Ez fázisváltozásának periódusa.

A Holdnak nincs saját fénye, ezért csak a Nap felé néző oldala világos. Keringése során ebből hol többet, hol kevesebbet fordít felénk, aszerint, hogy pályája melyik részén jár.

Amikor együttállásban van a Nappal, akkor sötét oldalát mutatja felénk, és nem látjuk, ez az *újhold*. Ahogy továbbhalad, egyre több látszik világosabb oldalából. Egy héttel az újhold után már félholdat mutat, ez az *első negyed*. További egy hét múlva szembenállásba kerül a Nappal, ez a *holdtölte* vagy *telehold*. Megint egy hét múlva, *utolsó negyedkor* ismét félholdat látunk.

A Nap állandóan ugyanakkora részét világítja meg égi kísérőnknek, a fázis nagyságát az határozza meg, hogy mi mekkora hányadát látjuk ennek. Amikor Holdunk, a Földről nézve a Nap közelében tartózkodik, megvilágított részét nem láthatjuk, ekkor van újhold. Az esti láthatósága során napsütötte részéből egyre többet figyelhetünk meg, és amikor ennek fele válik láthatóvá, elérkezünk az első negyedhez. Teleholdkor a teljes megvilágított félgömböt láthatjuk, utolsó negyedkor, hajnalban pedig ismét csak a felét. Amikor a Hold relatíve közel látszik a Naphoz (első negyed előtt, utolsó negyed után), árnyékos oldalán megfigyelhetjük a Föld légköréről a Holdra vetülő, majd onnan ismét visszaverődő fényt, ezt nevezik *hamuszürke fénynek*.

A holdfázisokat a következő ábrán láthatjuk:



A következő Delphi program kiszámítja és megjeleníti a Hold főbb adatait:

- a Hold fázisát
- a Hold korát
- a Hold távolságát
- az ekliptika szélességét és hosszúságát
- a Hold csillagjegyet

A dátumokat Juliánusz-dátum alakban használjuk.

```

program hold;
{$APPTYPE CONSOLE}
uses
  SysUtils, Math;

//felkerekít 2 tizedesre
function round2(x: real): real;
begin
  Result := (round(100*x)/100.0);
end;

//normalizál a 0..1 intervallumra
function normalize(v: real): real;
begin
  v := v - floor(v);
  if v < 0 then v := v+1;
  Result := v;
end;

```

```

//kiszámítja a Hold pozícióját és fázisát
procedure MoonPosit(Y, M, D: integer);
var
  AG: real;    //A Hold kora
  DI: real;    //A Hold távolsága Föld sugarakban
  LA: real;    //szélesség
  LO: real;    //hosszúság
  Phase: string;
  Zodiac: string;
  YY, MM, K1, K2, K3, JD: integer;
  IP, DP, NP, RP: real;

begin
  //kiszámítja a Juliánusz-dátumot 12 óra UT-kor
  YY := Y - floor((12 - M )/10);
  MM := M + 9;
  if (MM >= 12) then
    MM := MM - 12;
  K1 := floor(365.25 * (YY + 4712));
  K2 := floor(30.6 * MM + 0.5);
  K3 := floor(floor((YY / 100 ) + 49) * 0.75) - 38;

  JD := K1 + K2 + D + 59;    //Juliánusz-naptárbeli dátumokra
  if (JD > 2299160) then
    JD := JD - K3;          //Gergely-naptárbeli dátumokra

  //kiszámítja a Hold korát napokban
  IP := normalize((JD - 2451550.1) / 29.530588853);
  AG := IP*29.53;

  if AG < 1.84566 then Phase := 'újhold'
  else
    if AG < 5.53699 then Phase := 'holdsarló'
    else
      if AG < 9.22831 then Phase := 'első negyed'
      else
        if AG < 12.91963 then Phase := 'telő hold'
        else
          if AG < 16.61096 then Phase := 'telehold'
          else
            if AG < 20.30228 then Phase := 'fogyó hold'
            else
              if AG < 23.99361 then Phase := 'utolsó negyed'
              else
                if AG < 27.68493 then Phase := 'fekete hold'
                else
                  Phase := 'újhold';
  IP := IP*2*Pi;    //átalakítja radiánná

  //kiszámítja a Hold távolságát
  DP := 2*Pi*normalize((JD - 2451562.2)/27.55454988);
  DI := 60.4 - 3.3*cos(DP) - 0.6*cos(2*IP - DP) - 0.5*cos(2*IP);

  //kiszámítja a szélességet
  NP := 2*Pi*normalize((JD - 2451565.2) / 27.212220817);
  LA := 5.1*sin(NP);

```

```

//kiszámítja a hosszúságot
RP := normalize((JD - 2451555.8) / 27.321582241);
LO := 360*RP + 6.3*sin(DP) + 1.3*sin(2*IP - DP) + 0.7*sin(2*IP);

//meghatározzuk a Hold csillagjegyét
if LO < 33.18 then Zodiac := 'Pisces'
else
  if LO < 51.16 then Zodiac := 'Aries'
  else
    if LO < 93.44 then Zodiac := 'Taurus'
    else
      if LO < 119.48 then Zodiac := 'Gemini'
      else
        if LO < 135.30 then Zodiac := 'Cancer'
        else
          if LO < 173.34 then Zodiac := 'Leo'
          else
            if LO < 224.17 then Zodiac := 'Virgo'
            else
              if LO < 242.57 then Zodiac := 'Libra'
              else
                if LO < 271.26 then Zodiac := 'Scorpio'
                else
                  if LO < 302.49 then Zodiac := 'Sagittarius'
                  else
                    if LO < 311.72 then Zodiac := 'Capricorn'
                    else
                      if LO < 348.58 then Zodiac := 'Aquarius'
                      else Zodiac := 'Pisces';

//kiírjuk az eredményeket
writeln('Julian      = ', JD);
writeln('phase      = ', Phase);
writeln('age        = ', round2(AG):8:2, ' days');
writeln('distance   = ', round2(DI):8:2, ' earth radii');
writeln('ecliptic');
writeln(' latitude   = ', round2(LA):8:2, '°');
writeln(' longitude  = ', round2(LO):8:2, '°');
writeln('constellation = ', Zodiac);
end;

var
  year, month, day: integer;
begin
  write('Year: ');
  readln(year);
  write('Month: ');
  readln(month);
  write('Day: ');
  readln(day);
  writeln('Moon on ', month, '/', day, '/', year);
  MoonPosit(year, month, day);
  readln;
end.

```

Kovács Lehel István